

教育用擬似マイクロデータの符号表から

SAS データセット用の変数ラベルと変数フォーマットを自動作成する SAS プログラム

(公財) 統計情報研究開発センター 周防 節雄

要旨

日本の政府機関が作成したマイクロデータには、メタ情報として「符号表」が付随している。この符号表は、政府機関共通の「政府統計個票データレイアウト標準記法」(平成 18 年)に準拠して作成されており、エクセルファイルでネットからもアクセス・保存が出来る。そこには、変数名、変数の長さ、ラベル、変数コード値とその内容等が階層構造で表現されている。本報告では、統計センター提供の全国消費実態調査から作成された教育用擬似マイクロデータを SAS データセットに取り込んだ後、符号表から SAS 変数ラベルと変数フォーマットを自動生成し、かつ、SAS 変数も自動的に rename する SAS プログラムについて解説する。なお、この手法は他の政府統計のマイクロデータにも応用できる。

1. はじめに

近年、パソコンの性能の飛躍的な向上のおかげで、一昔前ならメインフレームコンピュータを駆使しないとできなかった大容量の公的マイクロデータの分析処理が大学の研究室内で可能となった。こうしたマイクロデータの「分析」手法は相当進んできているが、その元データをデータベースに編成して研究者が便利に利用できるようにする試みはあまり見られない。現状では、個々の研究者がそれぞれその都度苦勞してデータベースを構築しているのが現実である。

公的マイクロデータの提供を受けると、マイクロデータ本体の他に、メタ情報として「符号表」が付随してくる。この符号表は、政府機関共通の「政府統計個票データレイアウト標準記法」(平成 18 年)に準拠して作成されており、エクセルファイルでネットからもアクセス・保存が出来る。そこには、変数名、変数の長さ、ラベル、変数コード値とその内容等が階層構造で表現されている。通常、公的マイクロデータ本体は、固定長でカラム指定のテキストファイルか、コンマ区切りの CSV ファイルで提供される。各利用者は、符号表を参照しながらマイクロデータ本体を解釈して、必要な変数を取り出し分析処理に利用している。こうした処理は、変数の数やデータ件数が少ない場合はそれほど煩雑ではないが、統計法第 33 条に基づく調査票情報の提供(旧統計法の「目的外使用」に相当)で入手した公的マイクロデータや匿名データでは、変数の数やデータ件数が膨大になり、通常の手作業ではかなり手間と時間がかかり、仮にできたとしても、作業過程での手違いの可能性もあり得る。更に、一回分の調査データだけでなく、数回分のマイクロデータを利用する場合にはかなりの作業量になる。

著者自身の経験から、このデータベース編成の作業過程をシステム化することで、分析処理に供するデータベースの信頼性を高めたいと長年考えてきた。マイクロデータ提供時に入手する符号表のデザインはかなり精査されており、これを入力データとして利用して、提供されたオリジナルのマイクロデータ本体を読み込むプログラムを作成して、データベースに編成する試みをここ数年行ってきた。この手法を使って、現在、匿名データや調査票情報の提供で入手した公的マイクロデータのデータベース編成も行なった結果、これまでの煩雑な作業の機械化に成功した。本報告では、全国消費実態調査から作成された教育用擬似マイクロデータを取りあげて、その手法の紹介をする。本論文で使用する擬似マイクロデータ(平成 16 年全国消費実態調査に基づく)は、(独立行政法人)統計センターに利用申請をすれば、誰でも無料で提供されるので、本システムのプログラムを実際に使って試用することができる。

2. 擬似マイクロデータの概要

以下に擬似マイクロデータの本体の最初のレコード(図 1)と符号表(図 2)を示す。

1, 2, 1, 1, 2, 1, 1, 2, 1, 1, 3, VV, 1, 18, 9, 2873, 0301, 370456, 9355, 215046, 9113, 211011, 1511, 194775, 9545, 0, 0, 0, 0, 0, 16235, 19656, 0, 16235, 19656, 16235, 19656, 0, 0, 4035, 76027, 2419, 53341, 1616, 22786, 131041, 8992, 1234, 32, 386, 0, 0, 0, 0, 0, 0, 7609, 51313, 0, 0, 24358, 12501, 370446, 9355, 138544, 4475, 110290, 0792, 37941, 54346, 3686, 76725, 902, 97426, 1535, 22605, 1060, 4607, 188, 13624, 1465, 15969, 922, 4658, 205, 95685, 206, 53744, 130, 19959, 2125, 90599, 1859, 22454, 266, 68145, 1244, 73895, 566, 30396, 506, 90943, 171, 52556, 4650, 22894, 3096, 20692, 494, 02047, 599, 28045, 460, 7211, 1175, 32196, 1114, 01883, 61, 30313, 1570, 67274, 178, 64251, 1392, 03022, 2118, 83267, 3120, 73433, 1439, 03168, 1681, 70265, 1345, 57649, 414, 59796, 175, 91771, 755, 06082, 526, 83885, 14910, 7656, 14910, 7656, 0, 0, 0, 0, 0, 19694, 34402, 14166, 65895, 742, 04843, 776, 55246, 4009, 08507, 4556, 983, 15, 1603, 30798, 1010, 95282, 0, 592, 35517, 946, 43806, 278, 49531, 846, 7985, 881, 94329, 0, 2742, 17449, 0, 1833, 1206, 129, 40788, 358, 35029, 1345, 36642, 315, 17477, 35, 9687, 279, 20607, 0, 188, 07758, 23, 81711, 118, 22933, 46, 03114, 164, 16139, 187, 91704, 53, 7231, 0, 11271, 73799, 1188, 87066, 0, 1431, 93401, 8650, 93332, 17052, 56238, 271, 55017, 6410, 20706, 0, 0, 6410, 20706, 10370, 80515, 0, 0, 0, 0, 13576, 75265, 8720, 8525, 2376, 48222, 1816, 510, 95, 662, 89798, 0, 0, 0, 662, 89798, 3453, 98017, 802, 08152, 125, 19983, 255, 80339, 41, 22403, 77, 74321, 302, 11106, 928, 88116, 1723, 01749, 291, 84992, 0, 0, 14, 67644, 0, 1383, 33634, 33, 15478, 0, 28254, 36828, 5988, 87187, 3832, 80497, 2156, 06689, 0, 20332, 05666, 11770, 9816, 7346, 69939, 0, 1214, 37568, 1933, 43975, 210312, 4478, 182574, 4127, 1530, 37051, 0, 1570, 37051, 0, 0, 0, 16766, 34204, 0, 8316, 88743, 0, 1084, 43513, 21590, 04022

図1 擬似マイクロデータの最初のレコード
 (但し、例示目的なので一部の値は変更済み)

政府統計コード	実施時期	平成16年	作成日	訂正日
統計調査名	全国消費実態調査	集計区分	2人以上勤労世帯	ファイル名
調査票名	擬似マイクロデータ(大規模データ)	(備考・補足事項)		(照会先等)
コード体系	Shift_JIS	「△」はブランク(空白)を示す。		
レコード長	2974			

行番号	項目名	階層	位置	バイト数	配置	型	種別	変数名	符号	符号内容	備考
1	世帯区分	1	1	1				SetaiKubun	1	勤労	
2									2	勤労以外	
3									3	無職	
4		1	2	1					,	項目の区切り	
5	世帯員	1									
6	世帯人員	2	3	2	2	1		SetaiJinin	△2~	2人~	
7		2	5	1						項目の区切り	
8	有業人員	2	6	2	2	1		ShuugyouJinin	△1~	1人~	
9									VV	不詳	
10		2	8	1					,	項目の区切り	
11	現住居等に関する事項	2									
12	住居の構造	3	9	1				Kouzou	1	木造	
13									2	防火木造	
14									3	鉄骨・鉄筋コンクリート造	
15									4	その他(ブロック造り、レンガ造りなど)	
16									V	不詳	
17		3	10	1					,	項目の区切り	
18	住居の建て方	3	11	1				Tatekata	1	一戸建	
19									2	長屋建	
20									3	共同住宅(1・2階建)	
21									4	共同住宅(3~5階建)	
22									5	共同住宅(6~10階建)	
23									6	共同住宅(11階建以上)	
24									7	その他	
25									V	不詳	
26		3	12	1					,	項目の区切り	
27	住居の所有関係	3	13	1				Shoyuu	1	持ち家(世帯員名義)	
28									2	持ち家(その他名義)	
29									3	民営賃貸住宅(設備専用)	
30									4	民営賃貸住宅(設備共用)	
31									5	県市区町村営賃貸住宅	
32									6	都市再生機構・公社等賃貸住宅	
33									7	社宅・公務員住宅(借上げ含む)	
34									8	借間	
35									9	寮・寄宿舎	
36									V	不詳	
37		2	14	1					,	項目の区切り	
38	世帯員に関する事項	2									
39	世帯主	3									
40	性別	4	15	1				S1_Sex	1	男	
41									2	女	

中略

512	有価証券購入	4									
513	株式購入	5	2848	15	2	1		Youto176	△△△△△	円	
514		5	2863	1					,	項目の区切り	
515	他の有価証券購入	5	2864	15	2	1		Youto177	△△△△△	円	
516		5	2879	1					,	項目の区切り	
517	土地家屋借入金返済	4	2880	15	2	1		Youto178	△△△△△	円	
518		4	2895	1					,	項目の区切り	
519	他の借入金返済	4	2896	15	2	1		Youto179	△△△△△	円	
520		4	2911	1					,	項目の区切り	
521	分割払一括払購入借入金返済	4	2912	15	2	1		Youto180	△△△△△	円	
522		4	2927	1					,	項目の区切り	
523	財産購入	4	2928	15	2	1		Youto181	△△△△△	円	
524		4	2943	1					,	項目の区切り	
525	その他	4	2944	15	2	1		Youto182	△△△△△	円	
526		4	2959	1					,	項目の区切り	
527	繰越金	3	2960	15	2	1		Youto183	△△△△△	円	

図2 擬似マイクロデータ用符号表

統計センターのホームページに図3に示すマイクロデータの概要がある。本論文では「大規模データ」を使用しており、197項目から成る32027レコードのファイルがCSV形式のテキストファイルで7つのファイルに分割されている。

以下で解説するSASプログラムは、この7つのCSVファイルと図2のエクセルファイルを入力データとしている。

○全国消費実態調査

年次	調査の概要	データレイアウト及び符号表等	レコード数・項目数	擬似マイクロデータ
平成16年 (二人以上の世帯・勤労世帯)	調査ページ (総務省統計局)	大規模データ (XLS)	約3万2千 197項目	CSV形式 (ZIP (パスワード付)) 00001-05000 *1: 約4MB 05001-10000 : 約4MB 10001-15000 : 約4MB 15001-20000 : 約4MB 20001-25000 : 約4MB 25001-30000 : 約4MB 30001-32027 : 約2MB
		簡易データ ・CSV形式(XLS) ・EXCEL形式(XLS)	約8千 25項目	CSV形式 *1 (ZIP (パスワード付)): 約1MB EXCEL形式 *1,2 (ZIP (パスワード付)): 約1MB

図3 擬似マイクロデータ概要

(統計センターのホームページから転載)

3. システムの概要

3.1 SAS データセット作成プログラムの特徴

- (1) エンドユーザのパソコン環境に合わせて、提供されたマイクロデータの元ファイルが保存されているフォルダのパスを設定するだけで利用できる。ファイル名を指定する必要はない。従って、多数の元ファイルがある場合には、便利である。
- (2) 全ての変数を一旦「文字型」変数 X1、X2、…、X197 として取り込んだ後、数値型に変換する。
これは、半角アルファベットの「V」と「VV」がコード値として使用されていることの対策である。
- (3) 変数 X1～X197 を、「符号表」にある 変数名にすべて自動的に rename して、SAS データセットを作成する。(ただし、図2で□で囲った変数 Youto001～Youto183 は Y1～Y183 に rename した。)
- (4) 符号表にある日本語表記の「項目名」を使って 全ての変数に変数ラベルを自動設定 する。その結果、データ分析の段階で出力する図表に、変数名ではなくて変数ラベルで表示できるようになる。
- (5) コード値で表現されている変数を使って図表を出力する際、コード値ではなくて日本語表記ができるように、図2の符号表にある「変数名」、「符号」、「符号内容」を使って、SAS フォーマットの作成を自動的に行う。フォーマット名としては、変数名の末尾に「X」を付した(付録プログラム⑤参照)。こうすることでフォーマットを使用する際にフォーマット名を確認する手間が省ける。また、国勢調査のマイクロデータのように、符号表にある変数名の中には、数字で終わる変数名があり、そのままではフォーマット名としては使えない場合でも、この命名法であれば問題はない。

3.2 プログラムの構造と機能

本システムは次の4つのSASプログラムから成り、末尾の付録プログラム①～④に収録している。

- ① read_giji_zensho(all_at_once).sas
- ② giji_zensho_codetable_import.sas
- ③ label_rename.sas
- ④ create_format.sas

これらのプログラムの相互関係は図4に示す。この①から④までをこの順番で実行すれば、複数のCSVファイルで提供された擬似マイクロデータをSASデータセットとして保存できる。このデータセットの変数は、符号表にある「変数名」で命名されており、かつ、符号表にある「項目名」を使って変数ラベルが定義されている。更に、符号表の「符号」と「符号内容」を使って、変数フォーマットを定義するSASプログラム⑤(本稿末尾に収録)が自動作成される。




図4 プログラムの構造

3.3 本システムの利用方法

本システムを使用する際の留意点が2つある。

一つ目は、**図2の符号表の「行番号」、「項目名」・・・の行から上の行を全て削除しておく**ことである。その際、オリジナルのエクセルシートの複製のシートを作って、そのシート上で削除して使用することを勧める。

二つ目としては、統計センター提供の擬似マイクロデータと符号表を保存しているドライブ名やフォルダ名の設定が必要になる。そのために、**プログラム①と②の冒頭部にある%let文を使ってSASマクロ変数をそれぞれ5つと、2つ指定しなければならない。**

G:¥全消¥全消擬似マイクロデータ	
¥ マイクロデータcsv	
GIJI_2004zensho_dataset(00001~05000).csv	
GIJI_2004zensho_dataset(05001~10000).csv	
GIJI_2004zensho_dataset(10001~15000).csv	
GIJI_2004zensho_dataset(15001~20000).csv	
GIJI_2004zensho_dataset(20001~25000).csv	
GIJI_2004zensho_dataset(25001~30000).csv	
GIJI_2004zensho_dataset(30001~32027).csv	
¥ zensho_permanent	
zensho.sas7bdat	←自動作成される
code_table.sas7bdat	←自動作成される
¥ 統計センター_layout_code(鍵なし).xlsx	
	←符号表
¥ 統数研研究会用program	
read_giji_zensho(all_at_once).sas	←プログラム①
giji_zensho_codetable_import.sas	←プログラム②
label_rename.sas	←プログラム③
create_format.sas	←プログラム④
giji_micro_format.sas	←自動作成される

図5 筆者の利用環境

本稿の末尾の付録に示したSASプログラムは、著者が利用しているパソコン環境の設定なので、エンドユーザは各自の環境に合わせた設定が必要になる。ちなみに、著者の環境は図5に示す通りである。白抜き文字のファイルは本システムを実行すれば、自動作成される。

以上のマクロ変数の設定をした後に、プログラム①から④までを順番に実行すれば良い。その結果、擬似マイクロデータはzenshoという名前のパーマネントSASデータセットとして作成され、かつ、変数フォーマットを定義するSASプログラム「giji_micro_format.sas」(付録プログラム⑤)も作成される。

ここで解説した全てのSASプログラムは以下のURLから無料でダウンロードできる。プログラムの実行に当たっては、エンドユーザは、各自で擬似マイクロデータと符号表を入手されたい。

http://mighty.gk.u-hyogo.ac.jp/confidential/giji_micro_SAS_program.zip

なお、プログラム①と②で必要なマクロ変数を設定した後であれば、右のプログラム「だけ」を実行しても、プログラム①から④までが順番に自動的に実行されて、同じ結果が得られる。ただし、**このプログラムの冒頭部で、プログラム①から④までが保存されているフォルダのパスの設定が必要**となる。図5を参考にされたい。なお、このプログラムはどこに保存していても正常に作動する。

```

/* create_SAS_dataset.sas */
*このプログラムを実行すれば、マイクロデータ本体がSASデータセットに取り込まれ、かつ、符号表にある変数名と変数ラベルがセットされる。;

*★ 以下の①～④のSASプログラムが保存されているパスを指定して下さい;
%let program_path=G:¥全消¥全消擬似マイクロデータ¥統数研研究会用program;

*① マイクロデータ本体をSASデータセットに取り込む;
%include "&program_path¥read_giji_zensho(all_at_once).sas";

*② 符号表をSASデータセットに変換;
%include "&program_path¥giji_zensho_codetable_import.sas";

*③ 教育用擬似マイクロデータの変数ラベル作成と変数のrename;
%include "&program_path¥label_rename.sas";

*④ proc formatを作成する;
%include "&program_path¥create_format.sas";
    
```

一括実行するプログラム

4. まとめ

本稿では、教育用擬似マイクロデータを SAS データセットに取り込む SAS プログラムを解説したが、ここで使用した手法は、匿名データやもっと膨大なマイクロデータの調査票情報の利用(例えば、国勢調査のマイクロデータ)にも応用できる。符号表のスタイルは、政府機関共通の「政府統計個票データレイアウト標準記法」(平成18年)に準拠して作成されているが、細部は統計調査が異なれば多様な形式が採用されているので、完全な汎用プログラムの開発は今の段階ではできない。例えば、筆者は現在、全国消費実態調査の1994年、1999年、2004年の匿名データを利用しているが、その符号表には変数名がないので、本稿で紹介したプログラムはそのままでは使えない。ただし、一部修正するだけで、SAS データセットを作成したり、変数ラベルや変数フォーマットを簡単に自動設定することはできる。数年前に利用した国勢調査のマイクロデータの場合、符号表には変数名があり、今回とほぼ同じ手法で SAS データセットに編成することができた。規模の大きいマイクロデータで、しかも複数の回次のデータを扱う場合は、大幅な作業の軽減をはかることができる。

謝辞

擬似マイクロデータを提供して頂いた(独立行政法人)統計センターに対し、厚く御礼申し上げます。

参考文献

周防節雄(2014) 政府統計マイクロデータの符号表から SAS 変数のラベルとフォーマットを自動生成する SAS プログラムの作成方法、『SAS ユーザ総会 2014 論文集』、pp831-844

付録 SAS プログラム①

```
/* ① read_giji_zensho(all_at_once).sas */ options nocenter macrogen;

*以下の5つのマクロ変数を指定してください。;
%let path=G:\全消\全消擬似マイクロデータ; *「全消擬似マイクロデータ・プロジェクト」用path名;
%let ds_folder=zensho permanent; *パーマネントSASデータセット用フォルダ名;
%let csvFolder=マイクロデータcsv; *教育用擬似マイクロデータのCSVファイルが保存されるフォルダ名;
%let dsname=zensho; *マイクロデータから作成されるパーマネントSASデータセット名;
%let SurveyName=全国消費実態調査:教育用擬似マイクロデータ; *調査名:結果の表示時に使用される;

libname zensho "&path%\&ds_folder";

*全国消費実態調査:擬似マイクロデータの全てのCSVファイル名の指定(ワイルドカード使用);
filename FL "&path%\&csvFolder\*.csv";

data zensho.&dsname;
  length X1-X197 $15; *全ての変数を一旦文字型変数として取り込む;
  infile FL dsd LRECL=32767;
  input X1-X197;
run;

proc print data=zensho.&dsname (obs=25); title "①-1 &SurveyName (先頭10件)"; run;

data zensho.&dsname; set; drop i X1-X197; *文字型変数を数値型変数に変換;
  array X {197}; *文字型変数;
  do i=1 to 13; if index(X[i],"V") then X[i]="." ; *数値型の欠損値(不詳);
    if X[i]=" " then X[i]="999"; *数値型のコード(非..);
  end;

  array F {197} 8; *数値変数;
  do i=1 to 197; F[i]=X[i]; end; *文字型変数を数値型変数に変換;
run;

proc print data=zensho.&dsname (obs=25); title "①-2 &SurveyName (先頭10件)"; run;
```

結果ビューア画面

①-1 全国消費実態調査: 教育用擬似マイクロデータ (先頭10件) 文字型変数

OBS	X1	X2	X3	X4	X5	X6	X7	X8	X9	X10	X11	X12	X13	X14	X15	X16	X17	X18
1	1	2	1	1	1	1	1	2	1	1	3	VV	1	18.9	2873.0301	370446.9355	215046.9113	21
2	1	2	1	1	1	1	1	2	1	1	3	VV	1	15.43333	5055.3722	1134349.617	591755.2635	58
3	1	2	1	1	1	1	1	2	1	1	3	VV	1	15.43333	3337.9791	754758.1659	431302.1345	42
4	1	2	1	1	1	1	1	2	1	1	5	VV	1	21.03333	3264.1714	459078.5193	299831.2748	29
5	1	2	1	1	1	1	1	2	1	1	5	VV	1	8.93333	4282.6977	477208.2677	280483.9491	28
6	1	2	1	1	1	1	1	2	1	1	5	VV	1	8.93333	4009.378	427095.2424	311914.6126	31
7	1	2	1	1	1	1	1	3	1	1	2	VV	3	16.9	5853.0164	649738.9722	339553.2592	29
8	1	2	1	1	1	1	1	3	1	1	2	VV	3	9.93333	2207.2164	568988.0138	280663.0595	24
9	1	2	1	1	1	1	1	3	1	1	2	VV	3	10.96667	2248.3056	454047.5519	221530.6464	19
10	1	2	1	1	1	1	1	3	1	1	2	VV	3	14.1	3947.4113	453147.6286	285956.1176	24

①-2 全国消費実態調査: 教育用擬似マイクロデータ (先頭10件) 数値型変数

OBS	F1	F2	F3	F4	F5	F6	F7	F8	F9	F10	F11	F12	F13	F14	F15	F16	F17	F18
1	1	2	1	1	1	1	1	2	1	1	3	.	1	18.9000	2873.03	370446.94	215046.91	211011.15
2	1	2	1	1	1	1	1	2	1	1	3	.	1	15.4333	5055.37	1134349.62	591755.26	580649.86
3	1	2	1	1	1	1	1	2	1	1	3	.	1	15.4333	3337.98	754758.17	431302.13	423207.94
4	1	2	1	1	1	1	1	2	1	1	5	.	1	21.0333	3264.17	459078.52	299831.27	299831.27
5	1	2	1	1	1	1	1	2	1	1	5	.	1	8.9333	4282.70	477208.27	280483.95	280483.95
6	1	2	1	1	1	1	1	2	1	1	5	.	1	8.9333	4009.38	427095.24	311914.61	311914.61
7	1	2	1	1	1	1	1	3	1	1	2	.	3	16.9000	5853.02	649738.97	339553.26	295731.14
8	1	2	1	1	1	1	1	3	1	1	2	.	3	9.9333	2207.22	568988.01	280663.06	244441.20
9	1	2	1	1	1	1	1	3	1	1	2	.	3	10.9667	2248.31	454047.55	221530.65	192940.31
10	1	2	1	1	1	1	1	3	1	1	2	.	3	14.1000	3947.41	453147.63	285956.12	249051.14

付録 SAS プログラム②

```

/* ② giji_zensho_codetable_import.sas */ *符号表をSASデータセットに変換;

%let Codetable=統計センター_layout_code(鍵なし).xlsx; *符号表のファイル名;
%let sheet=SAS変数名なし; *符号表ファイルのシート名;

proc import out=zensho.code_table
datafile="&path%&CodeTable" dbms=excel replace; getnames=no; sheet="&sheet";
run;

data zensho.code_table; *項目名に対応する変数名にrenameする;
set;
rename F1=lineNO
       F2=Vlabel
       F3=strata
       F4=position
       F5=bytes
       F6=xhaichi
       F7=Vtype
       F8=xshubetsu
       F9=variable
       F10=Vcode
       F11=VcodeContents;
run;

proc print data=zensho.code_table; title "② &SurveyName.符号表"; run;
    
```

付録 SAS プログラム③

```

/* ③ label_rename.sas */ *教育用擬似マイクロデータの変数ラベル作成と変数のrename;
filename out1 "&path¥label.txt"; *変数ラベル定義用テキスト;
filename out2 "&path¥rename.txt"; *変数F1~F197のrename用テキスト;

* 全ての変数に変数ラベルを付するためのLABEL文の中味を外部ファイルに出力;
data label_define; keep var_label variable Vlabel;
  file out1; *LABEL文の中味を外部ファイルに出力;
  length var_label $ 100;
  set zensho.code_table; if variable="" then delete;

  *「Y」で始まる変数名を後で全て「Y」で始まる変数名(★数値型★)に置換するための
  rename文用変数名の作成;
  variable=tranwrd(variable,"Youto00","Y");
  variable=tranwrd(variable,"Youto0","Y");
  variable=tranwrd(variable,"Youto","Y");

  var_label=compress(variable) || '=' || compress(Vlabel || '');
  put var_label; *最終的に作成された全ての変数に付与するラベル文の作成;
run;
proc print; title "③-1 label"; var var_label variable Vlabel; run;

data _null_;
  file out2;
  *変数F1~F197を「符号表」の変数名に変えるrename文の中味をテキストファイルに出力;
  set label_define;
  statement=compress("F" || _N_ || "=" || variable);
  put statement;
run;

data zensho.&dsname; *変数F1~F197をrenameをする;
  set zensho.&dsname;
  rename
  %include "&path¥rename.txt"; ;
run;

proc print data=zensho.&dsname(obs=5); title "③-2 変数F1~F197のrename: (先頭5件)"; run;

data zensho.&dsname; *全ての変数にラベルを付ける; set label %include "&path¥label.txt"; ;run;
*マイクロデータのパーマネントデータセット完成;

proc print data=zensho.&dsname(obs=5);
  title "③-4 &SurveyName. パーマネントSASデータセット(先頭5件:変数名表示)"; run;
proc print data=zensho.&dsname(obs=5) label;
  title "③-5 &SurveyName. パーマネントSASデータセット(先頭5件 変数ラベル表示)"; run;

```

label.txt

```

SetaiKubun="世帯区分"
SetaiJinin="世帯人員"
ShuugyouJinin="有業人員"
Kouzou="住居の構造"
Tatekata="住居の建て方"
Shoyuu="住居の所有関係"
S1_Sex="性別"
S1_Age="年齢5歳階級"
S1_Shuugyou="就業・非就業の別"
S1_KigyokuKubun="企業区分"
S1_KigyokuKibo="企業規模"
S1_Sangyou="産業符号"
S1_Shokugyou="職業符号"
Weight="集計用乗率"
Y1="年間収入"
Y2="収入総額"
Y3="実収入"
Y4="経常収入"

```

以下省略

rename.txt

```

F1=SetaiKubun
F2=SetaiJinin
F3=ShuugyouJinin
F4=Kouzou
F5=Tatekata
F6=Shoyuu
F7=S1_Sex
F8=S1_Age
F9=S1_Shuugyou
F10=S1_KigyokuKubun
F11=S1_KigyokuKibo
F12=S1_Sangyou
F13=S1_Shokugyou
F14=Weight
F15=Y1
F16=Y2
F17=Y3
F18=Y4

```

以下省略

② 全国消費実態調査:教育用擬似マイクロデータ符号表

結果ビューア画面

OBS	lineNO	Vlabel	strata	position	bytes	xhaichi	Vtype	xshubetsu	variable	Vcode	VcodeContents
1	1	世帯区分	1	1	1	.	.	.	SetaiKubun	1	勤労
2	2			2	勤労以外
3	3			3	無職
4	4		1	2	1	項目の区切り
5	5	世帯票	1	
6	6	世帯人員	2	3	2	2	1	.	SetaiJinin	△2～	2人～
7	7		2	5	1	項目の区切り
8	8	有業人員	2	6	2	2	1	.	ShuugyouJinin	△1～	1人～
9	9			VV	不詳
10	10		2	8	1	項目の区切り
11	11	現住居等に関する事項	2	
12	12	住居の構造	3	9	1	.	.	.	Kouzou	1	木造
13	13			2	防火木造
14	14			3	鉄骨・鉄筋コンクリート造
15	15			4	その他(ブロック造り、レンガ造りなど)
16	16			V	不詳
17	17		3	10	1	項目の区切り
18	18	住居の建て方	3	11	1	.	.	.	Tatekata	1	一戸建
19	19			2	長屋建
20	20			3	共同住宅(1・2階建)
21	21			4	共同住宅(3～5階建)
22	22			5	共同住宅(6～10階建)
23	23			6	共同住宅(11階建以上)
24	24			7	その他
25	25			V	不詳

③-1 label

OBS	var_label	variable	Vlabel
1	SetaiKubun="世帯区分"	SetaiKubun	世帯区分
2	SetaiJinin="世帯人員"	SetaiJinin	世帯人員
3	ShuugyouJinin="有業人員"	ShuugyouJinin	有業人員
4	Kouzou="住居の構造"	Kouzou	住居の構造
5	Tatekata="住居の建て方"	Tatekata	住居の建て方
6	Shoyuu="住居の所有関係"	Shoyuu	住居の所有関係
7	S1_Sex="性別"	S1_Sex	性別
8	S1_Age="年齢5歳階級"	S1_Age	年齢5歳階級
9	S1_Shuugyou="就業・非就業の別"	S1_Shuugyou	就業・非就業の別
10	S1_KigyokuKubun="企業区分"	S1_KigyokuKubun	企業区分
11	S1_KigyokuKibo="企業規模"	S1_KigyokuKibo	企業規模
12	S1_Sangyou="産業符号"	S1_Sangyou	産業符号
13	S1_Shokugyou="職業符号"	S1_Shokugyou	職業符号
14	Weight="集計用乗率"	Weight	集計用乗率
15	Y1="年間収入"	Y1	年間収入
16	Y2="収入総額"	Y2	収入総額
17	Y3="実収入"	Y3	実収入
18	Y4="経常収入"	Y4	経常収入
19	Y5="勤め先収入"	Y5	勤め先収入
20	Y6="事業・内職収入"	Y6	事業・内職収入
21	Y7="農林漁業収入"	Y7	農林漁業収入
22	Y8="家賃収入"	Y8	家賃収入
23	Y9="他の事業収入"	Y9	他の事業収入
24	Y10="内職収入"	Y10	内職収入
25	Y11="本業以外の勤め先・事業・内職収入"	Y11	本業以外の勤め先・事業・内職収入

以下省略

㊦-2 変数F1~F197のrename: (先頭5件)

rename 後

OBS	SetaiKubun	SetaiJinin	ShuugyouJinin	Kouzou	Tatekata	Shoyuu	S1_Sex	S1_Age	S1_Shugyou	S1_KigyokuKubun	S1_KigyokuKibo	S1_Sangyou	S1_Shokugyou	Weight	Y1	Y2
1	1	2	1	1	1	1	1	2	1	1	3	.	1	18.9000	2873.03	370446.94
2	1	2	1	1	1	1	1	2	1	1	3	.	1	15.4333	5055.37	1134349.62
3	1	2	1	1	1	1	1	2	1	1	3	.	1	15.4333	3337.98	754758.17
4	1	2	1	1	1	1	1	2	1	1	5	.	1	21.0333	3264.17	459078.52
5	1	2	1	1	1	1	1	2	1	1	5	.	1	8.9333	4282.70	477208.27

㊦-4 全国消費実態調査: 教育用擬似マイクロデータ パーマネントSASデータセット(先頭5件 変数名表示)

変数ラベル付与後 (label 指定なし)

OBS	SetaiKubun	SetaiJinin	ShuugyouJinin	Kouzou	Tatekata	Shoyuu	S1_Sex	S1_Age	S1_Shugyou	S1_KigyokuKubun	S1_KigyokuKibo	S1_Sangyou	S1_Shokugyou	Weight	Y1	Y2
1	1	2	1	1	1	1	1	2	1	1	3	.	1	18.9000	2873.03	370446.94
2	1	2	1	1	1	1	1	2	1	1	3	.	1	15.4333	5055.37	1134349.62
3	1	2	1	1	1	1	1	2	1	1	3	.	1	15.4333	3337.98	754758.17
4	1	2	1	1	1	1	1	2	1	1	5	.	1	21.0333	3264.17	459078.52
5	1	2	1	1	1	1	1	2	1	1	5	.	1	8.9333	4282.70	477208.27

㊦-5 全国消費実態調査: 教育用擬似マイクロデータ パーマネントSASデータセット(先頭5件 変数ラベル表示)

変数ラベル付与後 (label 指定あり)

OBS	世帯区分	世帯人員	有業人員	住居の構造	住居の建て方	住居の所有関係	性別	年齢5歳階級	就業・非就業の別	企業区分	企業規模	産業符号	職業符号	集計用業率	年間収入	収入総額	実収入	経常収入	勤め先収入	農業・内職収入	農林漁業収入	家賃収入	他の事業収入	内職収入	本業以外の勤め先・事業・内職収入	他の経常収入	財産収入	社会保険給付	公的年金給付	他の年金給付	仕送り金	特別収入	受贈金		
1	1	2	1	1	1	1	1	2	1	1	3	.	1	18.9000	2873.03	370446.94	215046.91	211011.15	194775.95	0	0	0	0	0	0	0	16235.20	0	16235.20	16235.20	0	0	0	4035.76	2419.53
2	1	2	1	1	1	1	1	2	1	1	3	.	1	15.4333	5055.37	1134349.62	591755.26	580649.86	535974.67	0	0	0	0	0	0	0	44675.20	0	44675.20	44675.20	0	0	0	11105.40	6657.95
3	1	2	1	1	1	1	1	2	1	1	3	.	1	15.4333	3337.98	754758.17	431302.13	423207.94	390646.32	0	0	0	0	0	0	0	32561.62	0	32561.62	32561.62	0	0	0	8094.20	4652.66
4	1	2	1	1	1	1	1	2	1	1	5	.	1	21.0333	3264.17	459078.52	299831.27	299831.27	299831.27	0	0	0	0	0	0	0	0.00	0	0.00	0.00	0.00	0	0	0.00	0.00
5	1	2	1	1	1	1	1	2	1	1	5	.	1	8.9333	4282.70	477208.27	280483.95	280483.95	280483.95	0	0	0	0	0	0	0	0.00	0	0.00	0.00	0.00	0	0	0.00	0.00

変数ラベルと変数フォーマットの使用例

(この結果を出力したテストプログラムは本稿の末尾にある。)

全国消費実態調査: 教育用擬似マイクロデータ

FREQ プロシジャ

住居の構造				
Kouzou	度数	パーセント	累積度数	累積パーセント
不詳	5583	17.43	5583	17.43
木造	14064	43.91	19647	61.35
防火木造	3626	11.32	23273	72.67
鉄骨・鉄筋コンクリート造	8739	27.29	32012	99.95
その他(ブロック造り、レンガ造りなど)	15	0.05	32027	100.00

全国消費実態調査: 教育用擬似マイクロデータ

FREQ プロシジャ

企業規模				
S1_KigyokuKibo	度数	パーセント	累積度数	累積パーセント
不詳	2763	8.63	2763	8.63
1~4人	1095	3.42	3858	12.05
5~29人	5178	16.17	9036	28.21
30~499人	9334	29.14	18370	57.36
500~999人	1571	4.91	19941	62.26
1000人以上	5428	16.95	25369	79.21
非就業又は官公	6658	20.79	32027	100.00

全国消費実態調査: 教育用擬似マイクロデータ

FREQ プロシジャ

度数	表: Kouzou * S1_KigyokuKibo							
	S1_KigyokuKibo(企業規模)							
Kouzou(住居の構造)	不詳	1~4人	5~29人	30~499人	500~999人	1000人以上	非就業又は官公	合計
不詳	650	290	982	1418	372	857	1014	5583
木造	750	487	2491	4369	649	2026	3292	14064
防火木造	287	55	520	1120	144	612	888	3626
鉄骨・鉄筋コンクリート造	1073	263	1185	2421	406	1930	1461	8739
その他(ブロック造り、レンガ造りなど)	3	0	0	6	0	3	3	15
合計	2763	1095	5178	9334	1571	5428	6658	32027

付録 SAS プログラム④

```
/* ④ create_format.sas */ *マイクロデータ用のproc format作成;

filename out3 "&program_path¥giji_micro_format.sas";

data format; keep formatName Vcode VcodeContents;
  set zensho.code_table; *符号表のデータセットを使用;
  retain sw formatName;

  if Vlabel="," then sw=0;
  if variable NE " " AND Vtype=" " /*文字型変数*/
    then do; sw=1; formatName=variable; output; return; end;
  if sw=1 then output;
run;

proc print; title "④-1 format対象変数(1)"; var formatName Vcode VcodeContents; run;

*★===== dataset 'format' に対して、「V」と「△」の処理を追加する=====★;
data format; set format;
  if index(Vcode,"V") > 0 then Vcode=".";
  if (formatName="S1_KigyokuKibo" OR formatName="S1_KigyokuKubun") AND Vcode="△"
    then Vcode="999";
  else Vcode=kcompress(Vcode,"△");
run;
proc print; title "④-2 V converted to period"; run;
*★===== dataset 'format' に対する「V」と「△」の追加処理終了=====★;

proc sort data=format out=format1; by formatName; run;

data format1; set format1; by formatName;
  if first.formatName then statement2='value ' || compress(formatName || 'X' );
  statement=compress(Vcode || '=' || VcodeContents || '');
run;

proc print;
  title "④-3 format対象変数(2)";
  var formatName Vcode VcodeContents statement statement2;
run;

data _null_; file out3;
  set format1 end=final; by formatName;
  if _n_=1 then put "proc format; *&SurveyName;";
  if first.formatName then put statement2;
  if last.formatName then put " " statement ";" /;
  else put " " statement;
  if final then put "run;";
run;
```

結果ビューア画面

④-1 format対象変数(1)

OBS	formatName	Vcode	VcodeContents
1	SetaiKubun	1	勤労
2	SetaiKubun	2	勤労以外
3	SetaiKubun	3	無職
4	Kouzou	1	木造
5	Kouzou	2	防火木造
6	Kouzou	3	鉄骨・鉄筋コンクリート造
7	Kouzou	4	その他(ブロック造り、レンガ造りなど)
8	Kouzou	V	不詳
9	Tatekata	1	一戸建
10	Tatekata	2	長屋建
11	Tatekata	3	共同住宅(1・2階建)
12	Tatekata	4	共同住宅(3～5階建)
13	Tatekata	5	共同住宅(6～10階建)
14	Tatekata	6	共同住宅(11階建以上)
15	Tatekata	7	その他
16	Tatekata	V	不詳
17	Shoyuu	1	持ち家(世帯員名義)
18	Shoyuu	2	持ち家(その他名義)
19	Shoyuu	3	民営賃貸住宅(設備専用)
20	Shoyuu	4	民営賃貸住宅(設備共用)
21	Shoyuu	5	県市区町村営賃貸住宅
22	Shoyuu	6	都市再生機構・公社等賃貸住宅
23	Shoyuu	7	社宅・公務員住宅(借上げ含む)
24	Shoyuu	8	借間
25	Shoyuu	9	寮・寄宿舍
26	Shoyuu	V	不詳
27	S1_Sex	1	男
28	S1_Sex	2	女
29	S1_Age	△1	24歳未満
30	S1_Age	△2	25～29歳
31	S1_Age	△3	30～34歳
32	S1_Age	△4	35～39歳
33	S1_Age	△5	40～44歳
34	S1_Age	△6	45～49歳
35	S1_Age	△7	50～54歳
36	S1_Age	△8	55～59歳
37	S1_Age	△9	60～64歳
38	S1_Age	10	65～69歳
39	S1_Age	11	70～74歳
40	S1_Age	12	75歳以上
41	S1_Age	VV	不詳
42	S1_Shuugyou	1	就業
43	S1_Shuugyou	2	うちパート
44	S1_Shuugyou	3	非就業
45	S1_Shuugyou	4	うち仕事を探している
46	S1_Shuugyou	V	不詳

47	S1_KigyokuKubun	1	民営
48	S1_KigyokuKubun	2	自営
49	S1_KigyokuKubun	3	官公
50	S1_KigyokuKubun	△	非就業
51	S1_KigyokuKubun	V	不詳
52	S1_KigyokuKibo	1	1～4人
53	S1_KigyokuKibo	2	5～29人
54	S1_KigyokuKibo	3	30～499人
55	S1_KigyokuKibo	4	500～999人
56	S1_KigyokuKibo	5	1000人以上
57	S1_KigyokuKibo	△	非就業又は官公
58	S1_KigyokuKibo	V	不詳
59	S1_Sangyou	△1	農業
60	S1_Sangyou	△2	林業
61	S1_Sangyou	△3	漁業
62	S1_Sangyou	△4	鉱業
63	S1_Sangyou	△5	建設業
64	S1_Sangyou	△6	製造業
65	S1_Sangyou	△7	電気・ガス・熱供給・水道業
66	S1_Sangyou	△8	情報通信業
67	S1_Sangyou	△9	運輸業
68	S1_Sangyou	10	卸売・小売業
69	S1_Sangyou	11	金融・保険業
70	S1_Sangyou	12	不動産業
71	S1_Sangyou	13	飲食店・宿泊業
72	S1_Sangyou	14	医療・福祉
73	S1_Sangyou	15	教育・学習支援業
74	S1_Sangyou	16	複合サービス事業
75	S1_Sangyou	17	サービス業(他に分類されないもの)
76	S1_Sangyou	18	公務(他に分類されないもの)
77	S1_Sangyou	19	その他(非就業を含む)
78	S1_Sangyou	VV	不詳
79	S1_Shokugyou	△1	常用労務作業
80	S1_Shokugyou	△2	臨時及び日々雇労務作業
81	S1_Shokugyou	△3	民間職員
82	S1_Shokugyou	△4	官公職員1
83	S1_Shokugyou	△5	官公職員2
84	S1_Shokugyou	△6	商人及び職人
85	S1_Shokugyou	△7	個人経営者
86	S1_Shokugyou	△8	農林漁業従事者
87	S1_Shokugyou	△9	法人経営者
88	S1_Shokugyou	10	自由業者
89	S1_Shokugyou	11	その他
90	S1_Shokugyou	12	無職
91	S1_Shokugyou	VV	不詳

④-2 V converted to period

OBS	Vcode	VcodeContents	formatName
1	1	勤労	SetaiKubun
2	2	勤労以外	SetaiKubun
3	3	無職	SetaiKubun
4	1	木造	Kouzou
5	2	防火木造	Kouzou
6	3	鉄骨・鉄筋コンクリート造	Kouzou
7	4	その他(ブロック造り、レンガ造りなど)	Kouzou
8	.	不詳	Kouzou
9	1	一戸建	Tatekata
10	2	長屋建	Tatekata
11	3	共同住宅(1・2階建)	Tatekata
12	4	共同住宅(3～5階建)	Tatekata
13	5	共同住宅(6～10階建)	Tatekata
14	6	共同住宅(11階建以上)	Tatekata
15	7	その他	Tatekata
16	.	不詳	Tatekata
17	1	持ち家(世帯員名義)	Shoyuu
18	2	持ち家(その他名義)	Shoyuu
19	3	民営賃貸住宅(設備専用)	Shoyuu
20	4	民営賃貸住宅(設備共用)	Shoyuu
21	5	県市区町村営賃貸住宅	Shoyuu
22	6	都市再生機構・公社等賃貸住宅	Shoyuu
23	7	社宅・公務員住宅(借上げ含む)	Shoyuu
24	8	借間	Shoyuu
25	9	寮・寄宿舎	Shoyuu
26	.	不詳	Shoyuu
27	1	男	S1_Sex
28	2	女	S1_Sex
29	1	24歳未満	S1_Age
30	2	25～29歳	S1_Age
31	3	30～34歳	S1_Age
32	4	35～39歳	S1_Age
33	5	40～44歳	S1_Age
34	6	45～49歳	S1_Age
35	7	50～54歳	S1_Age
36	8	55～59歳	S1_Age
37	9	60～64歳	S1_Age
38	10	65～69歳	S1_Age
39	11	70～74歳	S1_Age
40	12	75歳以上	S1_Age
41	.	不詳	S1_Age
42	1	就業	S1_Shugyou
43	2	うちパート	S1_Shugyou
44	3	非就業	S1_Shugyou
45	4	うち仕事を探している	S1_Shugyou
46	.	不詳	S1_Shugyou

47	1	民営	S1_KigyokuKubun
48	2	自営	S1_KigyokuKubun
49	3	官公	S1_KigyokuKubun
50	999	非就業	S1_KigyokuKubun
51	.	不詳	S1_KigyokuKubun
52	1	1～4人	S1_KigyokuKibo
53	2	5～29人	S1_KigyokuKibo
54	3	30～499人	S1_KigyokuKibo
55	4	500～999人	S1_KigyokuKibo
56	5	1000人以上	S1_KigyokuKibo
57	999	非就業又は官公	S1_KigyokuKibo
58	.	不詳	S1_KigyokuKibo
59	1	農業	S1_Sangyou
60	2	林業	S1_Sangyou
61	3	漁業	S1_Sangyou
62	4	鉱業	S1_Sangyou
63	5	建設業	S1_Sangyou
64	6	製造業	S1_Sangyou
65	7	電気・ガス・熱供給・水道業	S1_Sangyou
66	8	情報通信業	S1_Sangyou
67	9	運輸業	S1_Sangyou
68	10	卸売・小売業	S1_Sangyou
69	11	金融・保険業	S1_Sangyou
70	12	不動産業	S1_Sangyou
71	13	飲食店・宿泊業	S1_Sangyou
72	14	医療・福祉	S1_Sangyou
73	15	教育・学習支援業	S1_Sangyou
74	16	複合サービス事業	S1_Sangyou
75	17	サービス業(他に分類されないもの)	S1_Sangyou
76	18	公務(他に分類されないもの)	S1_Sangyou
77	19	その他(非就業を含む)	S1_Sangyou
78	.	不詳	S1_Sangyou
79	1	常用労務作業	S1_Shokugyou
80	2	臨時及び日々雇労務作業	S1_Shokugyou
81	3	民間職員	S1_Shokugyou
82	4	官公職員1	S1_Shokugyou
83	5	官公職員2	S1_Shokugyou
84	6	商人及び職人	S1_Shokugyou
85	7	個人経営者	S1_Shokugyou
86	8	農林漁業従事者	S1_Shokugyou
87	9	法人経営者	S1_Shokugyou
88	10	自由業者	S1_Shokugyou
89	11	その他	S1_Shokugyou
90	12	無職	S1_Shokugyou
91	.	不詳	S1_Shokugyou

④-3 format対象変数(2)

OBS	formatName	Vcode	VcodeContents	statement	statement2
1	Kouzou	1	木造	1="木造"	value KouzouX
2	Kouzou	2	防火木造	2="防火木造"	
3	Kouzou	3	鉄骨・鉄筋コンクリート造	3="鉄骨・鉄筋コンクリート造"	
4	Kouzou	4	その他(ブロック造り、レンガ造りなど)	4="その他(ブロック造り、レンガ造りなど)"	
5	Kouzou	.	不詳	.="不詳"	
6	S1_Age	1	24歳未満	1="24歳未満"	value S1_AgeX
7	S1_Age	2	25～29歳	2="25～29歳"	
8	S1_Age	3	30～34歳	3="30～34歳"	
9	S1_Age	4	35～39歳	4="35～39歳"	
10	S1_Age	5	40～44歳	5="40～44歳"	
11	S1_Age	6	45～49歳	6="45～49歳"	
12	S1_Age	7	50～54歳	7="50～54歳"	
13	S1_Age	8	55～59歳	8="55～59歳"	
14	S1_Age	9	60～64歳	9="60～64歳"	
15	S1_Age	10	65～69歳	10="65～69歳"	
16	S1_Age	11	70～74歳	11="70～74歳"	
17	S1_Age	12	75歳以上	12="75歳以上"	
18	S1_Age	.	不詳	.="不詳"	
19	S1_KigyokuKibo	1	1～4人	1="1～4人"	value S1_KigyokuKiboX
20	S1_KigyokuKibo	2	5～29人	2="5～29人"	
21	S1_KigyokuKibo	3	30～499人	3="30～499人"	
22	S1_KigyokuKibo	4	500～999人	4="500～999人"	
23	S1_KigyokuKibo	5	1000人以上	5="1000人以上"	
24	S1_KigyokuKibo	999	非就業又は官公	999="非就業又は官公"	
25	S1_KigyokuKibo	.	不詳	.="不詳"	
26	S1_KigyokuKibun	1	民営	1="民営"	value S1_KigyokuKibunX
27	S1_KigyokuKibun	2	自営	2="自営"	
28	S1_KigyokuKibun	3	官公	3="官公"	
29	S1_KigyokuKibun	999	非就業	999="非就業"	
30	S1_KigyokuKibun	.	不詳	.="不詳"	
31	S1_Sangyou	1	農業	1="農業"	value S1_SangyouX
32	S1_Sangyou	2	林業	2="林業"	
33	S1_Sangyou	3	漁業	3="漁業"	
34	S1_Sangyou	4	鉱業	4="鉱業"	
35	S1_Sangyou	5	建設業	5="建設業"	
36	S1_Sangyou	6	製造業	6="製造業"	
37	S1_Sangyou	7	電気・ガス・熱供給・水道業	7="電気・ガス・熱供給・水道業"	
38	S1_Sangyou	8	情報通信業	8="情報通信業"	
39	S1_Sangyou	9	運輸業	9="運輸業"	
40	S1_Sangyou	10	卸売・小売業	10="卸売・小売業"	
41	S1_Sangyou	11	金融・保険業	11="金融・保険業"	
42	S1_Sangyou	12	不動産業	12="不動産業"	
43	S1_Sangyou	13	飲食店・宿泊業	13="飲食店・宿泊業"	
44	S1_Sangyou	14	医療・福祉	14="医療・福祉"	
45	S1_Sangyou	15	教育・学習支援業	15="教育・学習支援業"	
46	S1_Sangyou	16	複合サービス事業	16="複合サービス事業"	

47	S1_Sangyou	17	サービス業(他に分類されないもの)	17="サービス業(他に分類されないもの)"	
48	S1_Sangyou	18	公務(他に分類されないもの)	18="公務(他に分類されないもの)"	
49	S1_Sangyou	19	その他(非就業を含む)	19="その他(非就業を含む)"	
50	S1_Sangyou	.	不詳	.= "不詳"	
51	S1_Sex	1	男	1="男"	value S1_SexX
52	S1_Sex	2	女	2="女"	
53	S1_Shokugyou	1	常用労務作業者	1="常用労務作業者"	value S1_ShokugyouX
54	S1_Shokugyou	2	臨時及び日々雇労務作業者	2="臨時及び日々雇労務作業者"	
55	S1_Shokugyou	3	民間職員	3="民間職員"	
56	S1_Shokugyou	4	官公職員1	4="官公職員1"	
57	S1_Shokugyou	5	官公職員2	5="官公職員2"	
58	S1_Shokugyou	6	商人及び職人	6="商人及び職人"	
59	S1_Shokugyou	7	個人経営者	7="個人経営者"	
60	S1_Shokugyou	8	農林漁業従事者	8="農林漁業従事者"	
61	S1_Shokugyou	9	法人経営者	9="法人経営者"	
62	S1_Shokugyou	10	自由業者	10="自由業者"	
63	S1_Shokugyou	11	その他	11="その他"	
64	S1_Shokugyou	12	無職	12="無職"	
65	S1_Shokugyou	.	不詳	.= "不詳"	
66	S1_Shugyou	1	就業	1="就業"	value S1_ShugyouX
67	S1_Shugyou	2	うちパート	2="うちパート"	
68	S1_Shugyou	3	非就業	3="非就業"	
69	S1_Shugyou	4	うち仕事を探している	4="うち仕事を探している"	
70	S1_Shugyou	.	不詳	.= "不詳"	
71	Setaikubun	1	勤労	1="勤労"	value SetaikubunX
72	Setaikubun	2	勤労以外	2="勤労以外"	
73	Setaikubun	3	無職	3="無職"	
74	Shoyuu	1	持ち家(世帯員名義)	1="持ち家(世帯員名義)"	value ShoyuuX
75	Shoyuu	2	持ち家(その他名義)	2="持ち家(その他名義)"	
76	Shoyuu	3	民営賃貸住宅(設備専用)	3="民営賃貸住宅(設備専用)"	
77	Shoyuu	4	民営賃貸住宅(設備共用)	4="民営賃貸住宅(設備共用)"	
78	Shoyuu	5	県市区町村営賃貸住宅	5="県市区町村営賃貸住宅"	
79	Shoyuu	6	都市再生機構・公社等賃貸住宅	6="都市再生機構・公社等賃貸住宅"	
80	Shoyuu	7	社宅・公務員住宅(借上げ含む)	7="社宅・公務員住宅(借上げ含む)"	
81	Shoyuu	8	借間	8="借間"	
82	Shoyuu	9	寮・寄宿舎	9="寮・寄宿舎"	
83	Shoyuu	.	不詳	.= "不詳"	
84	Tatekata	1	一戸建	1="一戸建"	value TatekataX
85	Tatekata	2	長屋建	2="長屋建"	
86	Tatekata	3	共同住宅(1・2階建)	3="共同住宅(1・2階建)"	
87	Tatekata	4	共同住宅(3～5階建)	4="共同住宅(3～5階建)"	
88	Tatekata	5	共同住宅(6～10階建)	5="共同住宅(6～10階建)"	
89	Tatekata	6	共同住宅(11階建以上)	6="共同住宅(11階建以上)"	
90	Tatekata	7	その他	7="その他"	
91	Tatekata	.	不詳	.= "不詳"	

付録 プログラム⑤ 自動作成された SAS フォーマット (giji_micro_format.sas)

```

proc format; *全国消費実態調査：教育用擬似マイクロデータ;
value KouzouX
1="木造"
2="防火木造"
3="鉄骨・鉄筋コンクリート造"
4="その他（ブロック造り、レンガ造りなど）"
.="不詳";

value S1_AgeX
1="24歳未満"
2="25～29歳"
3="30～34歳"
4="35～39歳"
5="40～44歳"
6="45～49歳"
7="50～54歳"
8="55～59歳"
9="60～64歳"
10="65～69歳"
11="70～74歳"
12="75歳以上"
.="不詳";

value S1_KigyokuKiboX
1="1～4人"
2="5～9人"
3="10～19人"
4="20～49人"
5="50人以上"
999="非就業又は官公"
.="不詳";

value S1_KigyokuKubunX
1="民営"
2="自営"
3="官公"
999="非就業"
.="不詳";

value S1_SangyouX
1="農業"
2="林業"
3="漁業"
4="鉱業"
5="建設業"
6="製造業"
7="電気・ガス・熱供給・水道業"
8="情報通信業"
9="運輸業"
10="卸売・小売業"
11="金融・保険業"
12="不動産業"
13="飲食店・宿泊業"
14="医療・福祉"
15="教育・学習支援業"
16="複合サービス事業"
17="サービス業（他に分類されないもの）"
18="公務（他に分類されないもの）"
19="その他（非就業を含む）"
.="不詳";

value S1_SexX
1="男"
2="女";

value S1_ShokugyouX
1="常用労務作業者"
2="臨時及び日々雇労務作業者"
3="民間職員"
4="官公職員1"
5="官公職員2"
6="商人及び職人"
7="個人経営者"
8="農林漁業従事者"
9="法人経営者"
10="自由業者"
11="その他"
12="無職"
.="不詳";

value S1_ShuyouyouX
1="就業"
2="うちパート"
3="非就業"
4="うち仕事を探している"
.="不詳";

value SetaiKubunX
1="勤労"
2="勤労以外"
3="無職";

value ShoyuuX
1="持ち家（世帯員名義）"
2="持ち家（その他名義）"
3="民営賃貸住宅（設備専用）"
4="民営賃貸住宅（設備共用）"
5="県市区町村営賃貸住宅"
6="都市再生機構・公社等賃貸住宅"
7="社宅・公務員住宅（借上げ含む）"
8="借間"
9="寮・寄宿舎"
.="不詳";

value TatekataX
1="一戸建"
2="長屋建"
3="共同住宅（1・2階建）"
4="共同住宅（3～5階建）"
5="共同住宅（6～10階建）"
6="共同住宅（11階建以上）"
7="その他"
.="不詳";

run;
    
```

変数ラベルと変数フォーマットの使用例

（このプログラムの出力結果は、付録プログラム③の後ろにある。）

```

/* test_format.sas */ options nocenter;

%let path=G:\全消\全消擬似マイクロデータ; *「全消擬似マイクロデータ・プロジェクト」用path名;
%let ds_folder=zensho_permanent; *教育用擬似マイクロデータのCSVファイルが保存されるフォルダ名;
%let SurveyName=全国消費実態調査：教育用擬似マイクロデータ; *調査名：結果の表示時に使用される;

libname zensho "&path%\&ds_folder";
title "&SurveyName ";

proc freq data=zensho.zensho; tables kouzou / missing; format kouzou kouzouX.; run;

proc freq data=zensho.zensho; tables S1_KigyokuKibo / missing;
format S1_KigyokuKibo S1_KigyokuKiboX.; run;

proc freq data=zensho.zensho;
tables kouzou*S1_KigyokuKibo / norow nocol nopercnt missing;
format kouzou kouzouX.;
format S1_KigyokuKibo S1_KigyokuKiboX.;
run;
    
```